Darstellungsformen Funktionen mehrerer Veränderlicher (created with *Mathematica* 9)

Analytische Darstellung

In[1]:=
```
f = x^2 + y^2
```

Out[1]=
$$x^2 + y^2$$

In[2]:=
```
g = Sin[x * y]
```

Out[2]=
```
Sin[x y]
```

In[3]:=
```
p = T / V (* Konstante n R auf eins gesetzt! *)
```

Out[3]=
$$\frac{T}{V}$$

Funktionstafel

In[4]:=
```
Table[f, {x, 0, 3}, {y, 0, 3}] // MatrixForm
```

Out[4]//MatrixForm=
$$\begin{pmatrix} 0 & 1 & 4 & 9 \\ 1 & 2 & 5 & 10 \\ 4 & 5 & 8 & 13 \\ 9 & 10 & 13 & 18 \end{pmatrix}$$

Graphisch als Fläche im Raum

In[5]:=
```
Plot3D[f, {x, -3, 3}, {y, -3, 3}, AxesLabel → {"x", "y", "f"}]
```

Out[5]=



In[6]:=
```
Plot3D[f, {x, -3, 3}, {y, -3, 3}, AxesLabel → {"x", "y", "f"},
 RegionFunction → Function[{x, y, z}, x^2 + y^2 ≤ 9]]
```

Out[6]=



In[7]:=
```
Plot3D[g, {x, -3, 3}, {y, -3, 3},
 AxesLabel → {"x", "y", "g"}, Mesh → False, PlotPoints → 20]
```

Out[7]=



In[8]:=
```
Plot3D[p, {T, 0, 3}, {V, 0, 3},
 AxesLabel → {"T", "V", "p"}, PlotRange → {0, 20}, PlotPoints → 100]
```

Out[8]=



Höhenliniendiagramm

In[9]:=
```
ContourPlot[f, {x, -2, 2}, {y, -2, 2},
 Contours → {1, 2, 3, 4}, FrameLabel → {"x", "y"},
 ContourLabels → Function[{x, y, z}, Text[Framed[z], {x, y}, Background → White]]]
```

Out[9]=



In[10]:=
```
ContourPlot[g, {x, -3, 3}, {y, -3, 3}, Contours → Table[c, {c, -1, 1, 0.1}],
 FrameLabel → {"x", "y"}, ColorFunction → "BlueGreenYellow"]
```

Out[10]=



In[11]:=
```
ContourPlot[p, {T, 0, 3}, {V, 0, 3}, FrameLabel → {"T", "V"}, ContourLabels → True]
```

Out[11]=



Dichtediagramm

In[12]:=
```
DensityPlot[f, {x, -2, 2}, {y, -2, 2}, FrameLabel → {"x", "y"}]
```

Out[12]=



In[13]:=
```
DensityPlot[g, {x, -3, 3}, {y, -3, 3}, FrameLabel → {"x", "y"}, PlotPoints → 50]
```

Out[13]=



In[14]:=
```
DensityPlot[p, {T, 0, 3}, {V, 0, 3}, FrameLabel → {"T", "V"}, PlotPoints → 50]
```

Out[14]=



Schnittkurvendiagramm

In[15]:=
```
cuts = {0, 1, 2, 3};
```

In[16]:=
```
fc = f /. y -> cuts
```

Out[16]=
$$\left\{x^2,\ 1+x^2,\ 4+x^2,\ 9+x^2\right\}$$

In[17]:=
```
Plot[fc , {x, -3, 3},
 AxesLabel → {"x", "f"},
 PlotLegends → Map[("y=" <> ToString[#]) &, cuts], ImageSize → 400]
```

Out[17]=



In[18]:=
```
gc = g /. y -> cuts
```

Out[18]=
```
{0, Sin[x], Sin[2 x], Sin[3 x]}
```

In[19]:=
```
Plot[gc , {x, -3, 3},
 AxesLabel → {"x", "g"},
 PlotLegends → Map[("y=" <> ToString[#]) &, cuts], ImageSize → 400]
```

Out[19]=



In[20]:=
```
pc = p /. T -> cuts
```

Out[20]=
$$\left\{0, \frac{1}{v}, \frac{2}{v}, \frac{3}{v}\right\}$$

In[21]:=
```
Plot[pc , {V, 0, 3},
 AxesLabel → {"V", "p"}, PlotLegends → Map[("T=" <> ToString[#]) &, cuts],
 ImageSize → 400, AspectRatio → 1]
```

Out[21]=



Darstellung Funktionen dreier Veränderlicher (Beispiel: 3D ContourPlot)

In[22]:=
```
ContourPlot3D[x^2 + y^2 + z^2, {x, -3, 3}, {y, -3, 3}, {z, -3, 3}]
```

Out[22]=